

//Step01_Objects - prototype object

//VARIABLES

```
int division=2;
int xsize=600;
int ysize=600;
```

```
float xdivider=float(xsize)/float(division);
float ydivider=float(ysize)/float(division);
float xcoord=xdivider;
float ycoord=ydivider;
```

//SETUP

```
void setup()
{
    size(xsize,ysize);
    smooth();
}
```

//ANIMATION LOOP

```
void draw()
{
    background (27,27,77);

    // Drawing the needle

    float mouseangle=atan2(mouseX-xcoord,mouseY-ycoord);
    stroke(255);
    strokeWeight(2);

    pushMatrix();

    translate(xcoord, ycoord);
    rotate(-mouseangle);
    line(0, 0, 0, 250);

    popMatrix();
}
```

//Step02_Objects - creating object

```
//VARIABLES
int division=2;
int xsize=600;
int ysize=600;
float xdivider=float(xsize)/float(division);
float ydivider=float(ysize)/float(division);
float xcoord=xdivider;
float ycoord=ydivider;
Needle needle01;
```

```
//SETUP
```

```
void setup()
{
  size(xsize,ysize);
  needle01=new Needle();
  needle01.centreX=xcoord;
  needle01.centreY=ycoord;
  smooth();
}
```

```
//ANIMATION LOOP
```

```
void draw()
{
  background (27,27,77);
  needle01.display();
}
```

//Defining the Needle Class

```
class Needle {
```

```
  //VARIABLES
  float centreX;
  float centreY;
```

```
  //drawing function
```

```
  void display()
  {
    float mouseangle=atan2(mouseX-centreX,mouseY-centreY);
    stroke(255);
    strokeWeight(2);
    pushMatrix();
    translate(centreX, centreY);
    rotate(-mouseangle);
    line(0, 0, 0, 250);
```

```
    popMatrix();
  }
```

```
}
```

//Step03_Objects - adding further fields

```
//VARIABLES
int division=2;
int xsize=600;
int ysize=600;

float xdivider=float(xsize)/float(division);
float ydivider=float(ysize)/float(division);
float xcoord=xdivider;
float ycoord=ydivider;
color ncolor;
Needle needle01;

//SETUP

void setup()
{
  size(xsize,ysize);
  ncolor=color(248,252,15);
  needle01=new Needle();
  needle01.centreX=xcoord;
  needle01.centreY=ycoord;
  needle01.needle_length=100;
  needle01.needle_color=ncolor;
  needle01.needle_weight=2.5;
  smooth();
}

//ANIMATION LOOP
void draw()
{
  background (27,27,77);
  needle01.display();
}

//Defining the Needle Class

class Needle {

  //VARIABLES
  float centreX;
  float centreY;
  int needle_length;
  color needle_color;
  float needle_weight;

  //drawing function
  void display()
  {
    float mouseangle=atan2(mouseX-centreX,mouseY-centreY);
    stroke(needle_color);
    strokeWeight(needle_weight);

    pushMatrix();

    translate(centreX, centreY);
    rotate(-mouseangle);
    line(0, 0, 0, needle_length);

    popMatrix();
  }
}
```

//Step04_Objects - adding a constructor

```
//VARIABLES
int division=2;
int xsize=600;
int ysize=600;
float xdivider=float(xsize)/float(division);
float ydivider=float(ysize)/float(division);
float xcoord=xdivider;
float ycoord=ydivider;
color ncolor;
Needle needle01;

//SETUP
void setup()
{
  size(xsize,ysize);
  ncolor=color(248,252,15);
  // (x,y,length, color, weight) – creating object and calling constructor function
  needle01=new Needle(xcoord,ycoord,100,ncolor,2.5);
  smooth();
}

//ANIMATION LOOP
void draw()
{
  background (27,27,77);
  needle01.display();
}

//Defining the Needle Class
class Needle {
  //VARIABLES
  float centreX;
  float centreY;
  int needle_length;
  color needle_color;
  float needle_weight;

  //Needle constructor function
  // (x,y,length, color,weight)
  Needle(float needx, float needy, int needlength, color needcolor, float needweight)
  {
    centreX=needx;
    centreY=needy;
    needle_length=needlength;
    needle_color=needcolor;
    needle_weight=needweight;
  }

  //drawing function
  void display()
  {
    float mouseangle=atan2(mouseX-centreX,mouseY-centreY);
    stroke(needle_color);
    strokeWeight(needle_weight);
    pushMatrix();
    translate(centreX, centreY);
    rotate(-mouseangle);
    line(0, 0, 0, needle_length);
    popMatrix();
  }
}
```

//Step05_Objects - multiple objects

//VARIABLES

```
int division=4;  
int xsize=600;  
int ysize=600;  
float xdivider=float(xsize)/float(division);  
float ydivider=float(ysize)/float(division);  
float xcoord=xdivider;  
float ycoord=ydivider;  
color ncolor;  
color ncolor2;
```

```
Needle needle01;  
Needle needle02;
```

//SETUP

```
void setup()  
{  
  
    size(xsize,ysize);  
  
    ncolor=color(248,252,15);  
    ncolor2=color(255,0,0);  
  
    // (x,y,length, color, weight)  
    needle01=new Needle(xcoord,ycoord,150,ncolor,2.5);  
    needle02=new Needle(xcoord+xdivider,ycoord,100,ncolor2,2.5);  
  
    smooth();  
  
}
```

//ANIMATION LOOP

```
void draw()  
{  
    background (27,27,77);  
  
    needle01.display();  
    needle02.display();  
  
}
```

//Defining the Needle Class

// NOTE: CLASS DESCRIPTION REMAINS UNCHANGED FROM STEP 04

//Step06_Objects - using an array

```
//VARIABLES
```

```
int division=4;  
int xsize=600;  
int ysize=600;
```

```
float xdivider=float(xsize)/float(division);  
float ydivider=float(ysize)/float(division);  
float xcoord=xdivider;  
float ycoord=ydivider;  
color ncolor;  
//color ncolor2;
```

```
Needle[] needles = new Needle[2];
```

```
//Needle needle01;  
//Needle needle02;
```

```
//SETUP
```

```
void setup()  
{
```

```
    size(xsize,ysize);  
    ncolor=color(255,0,0);  
    //ncolor2=color(255,0,0);
```

```
    // (x,y,length, color, weight)
```

```
    needles[0]=new Needle(xcoord,ycoord,100,ncolor,2.5);  
    needles[1]=new Needle(xcoord+xdivider,ycoord,100,ncolor,2.5);
```

```
    smooth();
```

```
}
```

```
//ANIMATION LOOP
```

```
void draw()
```

```
{  
    background (27,27,77);
```

```
    needles[0].display();  
    needles[1].display();
```

```
}
```

```
//Defining the Needle Class
```

```
// NOTE: CLASS DESCRIPTION REMAINS UNCHANGED FROM STEP 04
```

//Step07_Objects - Forming a row - Populating the array

//VARIABLES

```
int division=5;
int xsize=600;
int ysize=600;
int nline=division-1;
```

```
float xdivider=float(xsize)/float(division);
float ydivider=float(ysize)/float(division);
//float xcoord=xdivider;
float[] xcoord = new float[nline];
//float ycoord=ydivider;
float[] ycoord = new float[nline];
```

```
Needle[]needles = new Needle[nline];
color ncolor;
```

//SETUP

```
void setup()
{
  size(xsize,ysize);
  ncolor=color(255,0,0);

  // (x,y,length, color, weight)
  // needles[0]=new Needle(xcoord,ycoord,100,ncolor,2.5);
  // needles[1]=new Needle(xcoord+xdivider,ycoord,100,ncolor,2.5);
```

```
  //populating the array
  for (int i=0; i<nline; i++)
  {
    xcoord[i]=xdivider*(i+1);
    ycoord[i]=ydivider;

    // (x,y,length, color, weight)
    needles[i]=new Needle(xcoord[i],ycoord[i],100,ncolor,1.5);
  }
```

```
  smooth();
```

```
}
```

//ANIMATION LOOP

```
void draw()
{
  background (27,27,77);

  // needles[0].display();
  // needles[1].display();
```

```
  //displaying all the needles in the array
  for (int i=0; i<nline; i++)
  {
    needles[i].display();
  }
}
```

//Defining the Needle Class

// NOTE: CLASS DESCRIPTION REMAINS UNCHANGED FROM STEP 04

//Step08_Objects - 2-dimensional arrays

//VARIABLES

```
int division=4;
int xsize=600;
int ysize=600;
int nline=division-1;
float xdivider=float(xsize)/float(division);
float ydivider=float(ysize)/float(division);
```

```
float[][]xcoord = new float[nline][nline];
float[][]ycoord = new float[nline][nline];
```

```
Needle[][]needles = new Needle[nline][nline];
color ncolor;
```

//SETUP

```
void setup()
{
  size(xsize,ysize);
  ncolor=color(255,0,0);

  //populating the array
  for (int i=0; i<nline; i++)
  {
    for (int j=0; j<nline; j++)
    {
      xcoord[i][j]=xdivider*(i+1);
      ycoord[i][j]=ydivider*(j+1);

      // (x,y,length, color, weight)
      needles[i][j]=new Needle(xcoord[i][j],ycoord[i][j],100,ncolor,1.5);
    }
  }

  smooth();
}
```

//ANIMATION LOOP

```
void draw()
{
  background (27,27,77);

  //displaying all the needles in the array
  for (int i=0; i<nline; i++)
  {
    for (int j=0; j<nline; j++)
    {
      needles[i][j].display();
    }
  }
}
```

//Defining the Needle Class

// NOTE: CLASS DESCRIPTION REMAINS UNCHANGED FROM STEP 04

//Step09_Objects - Coordinates as function

//VARIABLES

```
int division=20;
int xsize=600;
int ysize=600;
int nline=division-1;
```

```
float xdivider=float(xsize)/float(division);
float ydivider=float(ysize)/float(division);
float[][]xcoord = new float[nline][nline];
float[][]ycoord = new float[nline][nline];
```

```
Needle[][]needles = new Needle[nline][nline];
color ncolor;
```

//SETUP

```
void setup()
{
  size(xsize,ysize);
  ncolor=color(255,0,0);

  //populating the array
  for (int i=0; i<nline; i++)
  {
    for (int j=0; j<nline; j++)
    {
      xcoord[i][j]=xdivider*(i+1);
      ycoord[i][j]=ydivider*(j+1);

      // (x,y,length, color, weight)
      needles[i][j]=new Needle(xcoord[i][j],ycoord[i][j],100,ncolor,1.5);
    }
  }

  smooth();
}
```

//ANIMATION LOOP

```
void draw()
{
  background (27,27,77);

  //displaying all the needles in the array
  for (int i=0; i<nline; i++)
  {
    for (int j=0; j<nline; j++)
    {
      needles[i][j].display(mouseX,mouseY);
    }
  }
}
```

//Defining the Needle Class (see overleaf)

//Defining the Needle Class

```
class Needle {

    //VARIABLES
    float centreX;
    float centreY;
    int needle_length;
    color needle_color;
    float needle_weight;

    //Needle constructor function
    // (x,y,length, color,weight)

    Needle(float needx, float needy, int needlength, color needcolor, float needweight)
    {
        centreX=needx;
        centreY=needy;
        needle_length=needlength;
        needle_color=needcolor;
        needle_weight=needweight;
    }

    //drawing function

    void display(float targetX, float targetY)
    {
        float mouseangle=atan2(targetX-centreX,targetY-centreY);
        stroke(needle_color);
        strokeWeight(needle_weight);

        pushMatrix();

        translate(centreX, centreY);
        rotate(-mouseangle);
        line(0, 0, 0, needle_length);

        popMatrix();
    }
}
```